

**IBIS Interconnect SPICE Subcircuits Specification  
(IBIS-ISS)**

**Draft 0.9  
January 10, 2010**

Contents	
1	Overview ..... 5
2	Goals and Scope..... 6
3	Conventions ..... 7
4	Input Structure and Data Entry..... 8
4.1	Input File Guidelines ..... 8
4.2	Statements and Arguments..... 8
4.3	Special Characters..... 10
4.4	First Character ..... 13
4.5	Delimiters ..... 14
4.6	Instance Names ..... 14
4.7	Numbers..... 15
4.8	Parameters and Expressions ..... 16
4.9	Node Name (or Node Identifier) Conventions ..... 17
4.10	Element, Instance, and Subcircuit Naming Conventions ..... 17
5	Parameters ..... 19
5.1	Using Parameters in Simulation (.PARAM)..... 19
	Defining Parameters ..... 19
	Assigning Parameters ..... 20
5.2	Using Algebraic Expressions..... 21
	Built-In Functions and Variables..... 21
5.3	String Parameters ..... 26
5.4	Parameter Scoping and Passing..... 26
6	File Includes ..... 27
7	Comments..... 28
8	Model Definitions (.MODEL Statements) ..... 29
9	Subcircuit Definitions..... 30
9.1	Subcircuit Scoping Rules ..... 30
10	Subcircuit Definition Ending Statements..... 31
11	Elements ..... 32
11.1	Subcircuits ..... 32
11.2	Linear Resistor..... 32
11.3	Linear Capacitor..... 33
11.4	Voltage Source ..... 34
11.5	Mutual Inductor ..... 34

11.6	Linear Inductor .....	35
11.7	T-element (Ideal Transmission Line).....	35
11.8	W-element (Coupled Transmission Line).....	37
	Format 1: RLGC Model .....	38
	<b>Lgnd</b> .....	39
	Using RLGC Matrices .....	40
	Format 2: Frequency-Dependent Tabular Specification.....	40
11.9	Frequency-Dependent Matrices.....	41
	Small-Signal Parameter Data Frequency Table Model (SP Model) .....	42
11.10	S-element .....	46
	S-Element Model Syntax.....	48
11.11	E-element (Voltage-Controlled Voltage Source) .....	50
	Linear .....	50
	Laplace Transform .....	50
	Pole-Zero Function.....	50
	Foster Pole-Residue Form .....	51
	E-element Arguments .....	52
11.12	F-element (Current-Controlled Current Source).....	53
11.13	G-element (Voltage-Controlled Current Source).....	54
	Linear .....	54
	Laplace Transform .....	54
	Pole-Zero Function.....	54
	Foster Pole-Residue Form .....	55
	G-element Arguments .....	56
11.14	H-element (Current-Controlled Voltage Source) .....	56
12	Best Practices .....	58
13	Theoretical Background.....	59
	13.1 Introduction to the Complex Dielectric Loss Model .....	59
14	References .....	61

Table 1: Document Conventions .....	7
Table 2: Prohibited Operators for Specific Elements.....	10
Table 3: IBIS-ISS Special Characters .....	11
Table 4: First Character Rules .....	13
Table 5: Element Identifiers.....	14
Table 6: Scale Factors .....	15
Table 7: .PARAM Statement Syntax and Examples.....	19
Table 8: Parameter Passing Order.....	20
Table 9: IBIS-ISS Built-in Functions .....	21
Table 10: IBIS-ISS Special Variables .....	24
Table 11: File Include Arguments.....	27
Table 12: Subcircuit Arguments .....	32
Table 13: Linear Resistor Arguments .....	33
Table 14: Linear Capacitor Arguments.....	33
Table 15: Voltage Shunt Arguments .....	34
Table 16: Mutual Inductor Arguments .....	34
Table 17: Linear Inductor Arguments .....	35
Table 18: T-element (Ideal Transmission Line) Arguments.....	36
Table 19: W-element (Coupled Transmission Line) Arguments .....	37
Table 20: W-element RLC Model Arguments .....	39
Table 21: W-element SP Model Arguments .....	42
Table 22: W-element Model Definition Arguments .....	45
Table 23: S-element Arguments.....	46
Table 24: S-element Model Definition Arguments .....	48
Table 25: E-element Arguments.....	52
Table 26: F-element Arguments .....	53
Table 27: G-element Arguments .....	56
Table 28: H-element Arguments .....	57

# 1 Overview

The IBIS Open Forum, in order to enable easier data exchange between users of signal/power integrity simulation and physical layout/routing software tools, is issuing a generic netlist format, to be called “IBIS Interconnect SPICE Subcircuits” (IBIS-ISS).

This format is similar in structure and major functions to the SPICE (Simulation Program with Integrated Circuit Emphasis) nodal syntax developed at the University of California at Berkeley and since implemented in various forms by individual software tool vendors. IBIS-ISS is the first industry-wide attempt to standardize SPICE subcircuit representation.

This version of IBIS-ISS is based on a subset of HSPICE®, used with permission from Synopsys, Inc. HSPICE is a registered trademark of Synopsys, Inc.

## 2 Goals and Scope

The syntax of IBIS-ISS is intended to:

- describe interconnect structures (such as PCB traces, connectors, cables, etc.) electrically, for analysis in a signal integrity and/or power integrity context
- describe the arrangement or topology of interconnect structures, as they relate to each other and to active devices in a system

To these ends, IBIS-ISS includes support for:

- elementary circuit elements (resistors, capacitors, inductors)
- transmission line elements (lossless and lossy)
- frequency-domain network parameters (e.g., S-parameters)
- parameter/variable passing to elements and subcircuits
- dependent sources
- string-based node naming
- user-defined comments
- abstraction through modular, user-defined subcircuit definitions

IBIS-ISS does NOT include or cover:

- descriptions of complete netlists intended for input “as-is” to simulation tools
- model formats or “process cards” for active devices (e.g., diodes, transistors)
- independent sources
- controls or options for any simulation engine (e.g., precision, algorithm selection)
- simulation or analysis types (e.g., DC, transient)
- sweep or run control (e.g., Monte Carlo)
- geometrical descriptions for field solvers
- support for other kinds of data extraction/export (e.g., S-parameter generation)
- measurement, printing or probing
- encryption support

# Conventions

The following typographical conventions are used in IBIS-ISS. Note that these may be combined (e.g., Courier font in bold type).

Table 1: Document Conventions

Convention	Description
Courier	Indicates statement syntax
<i>Italic</i>	Indicates a user-defined value, where a specific text string will replace the italics shown in an actual IBIS-ISS file (e.g., <i>Rxxxx</i> is a generic representation of a resistor element name, such as <i>Rname</i> )
<b>Bold</b>	Indicates verbatim text in syntax descriptions
[ ]	Denotes optional arguments
...	Indicates that arguments of the same type may be added as appropriate to the element structure: pin1 pin2 ... pin
	Indicates a choice among defined alternatives, such as low   medium   high
+	Indicates a continuation of a statement across line termination sequence(s). Note that continuation may only be used between arguments and shall not split any single non-quoted-string argument across lines

## 4 Input Structure and Data Entry

This section describes the input file and structures for representing input data.

### 4.1 Input File Guidelines

An input file consists of a collection of statements describing a portion of a complete circuit. This input file is intended for inclusion in a larger netlist or description of a complete circuit, to be used by a simulation tool.

An input filename may be up to 1024 characters long. The input file shall be in ASCII format (as defined in ISO/IEC 8859-1:1998, with each character encoded in 8 bits). The input file shall not be in a binary, packed or compressed format.

### 4.2 Statements and Arguments

A statement in IBIS-ISS is a text string consisting of character sequence arguments and delimiters. An IBIS-ISS file may contain multiple statements (the number of statements is not limited by the IBIS-ISS definition, but may be limited by the computer architecture and/ or operating system used to process the file).

Any individual input line (combination of characters separated by line-termination sequences) may be up to 1024 characters long.

Statements in an input file may appear in any order.

Any valid string of characters between two delimiters is an argument.

For the purpose of this specification, statements are grouped into the following types:

- Element instances
- Parameter definitions
- File includes
- Subcircuit definitions
- Model definitions
- Comments
- Subcircuit ending statements

Subcircuit ending statements, subcircuit definitions, model definitions, parameter definitions and file includes all begin with the dot (.) character.

The specific syntax of the above statement types are described in the sections below.

- IBIS-ISS ignores differences between upper and lower case in input statements, except in quoted filenames.
- To continue a statement across multiple lines (i.e., across a line-termination sequence), the plus (+) sign shall be used as the first non-alphanumeric, non-blank character after the line-termination sequence. The + sign shall be used in addition to a delimiter and never to split up the character sequences between delimiters.
- Quoted strings with extended length (such as paths and expressions) may span a line-termination sequence using a double backslash ( \\ ) sequence at the end of the line containing the string to be continued on the following line. Note that the double backslash ( \\ ) sequence may be preceded by whitespace when used within a quoted string. Individual arguments may be split across line-termination sequences using a double backslash, but leading whitespace shall not be used.
- The following characters are reserved for special use and shall not be used as part of any parameter or node name:  
$$() = " '$$
- String expressions may be grouped by surrounding pairs of ‘ or “ characters.

- To prevent conflicts with proprietary SPICE variants, the following strings shall not be used as part of any parameter or node name in the associated element:

**Table 2: Prohibited Operators for Specific Elements**

Element	Prohibited String Sequence
Capacitor	POLY, TC, SENS
E-element G-element	AND, DELAY, FOSTER, LAPLACE, NAND, NPWL, NOR, VCCS, OPAMP, OR, POLE, POLY, PWL, SPUR, TRANSFORMER, VCR, VCCAP, VCVS, FREQ, ZTRANS, VMRF, NOISE, NOISEFILE, MNAME, PHASE, SCALE, MAX, PAR
F-element H-element	POLY, PWL, AND, NAND, OR, NOR, VMRF, CCCS, C CVS, DELAY
Inductor	POLY, TC, SENS, RELUCTANCE, TRANSFORMER_NT, FILE
Resistor	POLY,TC,SENS
S-element	ZO, Z0, MNAME
T-element	IC
W-element	RLGCFE, PRINTZO, RLGCMODEL, TABLEMODEL, FSMODEL, UMODEL, SMODEL

### 4.3 Special Characters

The following table lists the special characters that may be used as part of node names, element parameter names, and element instance names. For detailed discussion, see the appropriate sections in this chapter.

**Note:**

To avoid unexpected results or error messages, do not use the following mathematical characters in a parameter name in IBIS-ISS: \* - + ^ and /.

**Table 3: IBIS-ISS Special Characters**

Special Character*		Node Name	Instance Name**	Parameter Name**	Comments
~	Tilde	Legal anywhere	Included only	Included only	n/a
!	Exclamation point	Legal anywhere	Included only	Included only	n/a
@	At sign	Legal anywhere	Included only	Included only	n/a
#	Pound sign	Legal anywhere	Included only	Included only	n/a
\$	Dollar sign	Included only (avoid if after a number in node name)	Included only	Included only	In-line comment character
%	Percent	Legal anywhere	Included only	Included only	n/a
^	Caret	Legal anywhere	Included only	Included only (avoid usage)	“To the power of”, i.e., 2 <sup>5</sup> , two raised to the fifth power
&	Ampersand	Legal anywhere	Included only	Included only	n/a
*	Asterisk	Included only (avoid using in node names)	Included only	Included only (avoid using in parameter names)	Comment and wildcard character. Double asterisk (**) is “to the power of”.

( )	Parentheses	Illegal	Illegal	Illegal	Delimiter
-	Minus	Included only	Included only	Illegal	n/a
_	Underscore	Legal anywhere	Included only	Included only	n/a
+	Plus sign	Included only	Included only	Included only (avoid usage)	Continues previous line, except for quoted strings
=	Equals	Illegal	Illegal	Optional in .PARAM statements	Delimiter
< >	Less/more than	Legal anywhere	Included only	Included only	n/a
?	Question mark	Legal anywhere	Included only	Included only	Wildcard character
/	Forward slash	Legal anywhere	Included only	Illegal	n/a
{ }	Curly braces	included only, converts to [ ]	Included only	Included only	Engine shall auto-convert to square brackets ( [ ] )
[ ]	Square brackets	Included only	Included only	Included only	n/a
\	Double backslash	Included only	Illegal	Illegal	Continuation character sequence

	Pipe	Legal anywhere	Included only	Included only	n/a
,	Comma	Illegal	Illegal	Illegal	Delimiter
.	Period	Illegal	Included only	Included only	Statement identifier, (i.e., .PARAMETER, etc.).
:	Colon	Included only	Included only	Included only	Delimiter for element attributes
;	Semi-colon	Included only	Included only	Included only	n/a
" "	Double-quotes	Illegal	Illegal	Illegal	Expression and filename delimiter
' '	Single quotes	Illegal	Illegal	Illegal	Expression and filename delimiter
	Blank (whitespace)	May be used before \\ sequence for continuations	Illegal	Illegal	Delimiter

\*“Legal anywhere”=first character or any position in string

“Included only”=any position except first character

\*\* cannot be the first character; element key letter only

#### 4.4 First Character

The first character in every line specifies how IBIS-ISS interprets the remainder of the line.

Table 4: First Character Rules

If the First Character is...	Indicates
------------------------------	-----------

. (period)	Statement identifier (e.g., .PARAM)
c, C, e, E, f, F, g, G, h, H, k, K, l, L, r, R, s, S, t, T, v, V, w, W, x, X	Element instantiation
* (asterisk)	Comment line
+ (plus)	Continues previous line

## 4.5 Delimiters

Delimiters separate arguments in the input file. Input delimiters are: tab, blank, comma (,), equal sign (=), and parentheses ( ).

In addition, single (') or double quotes (") delimit and group expressions and filenames.

## 4.6 Instance Names

The names of element instances begin with the element key letter or identifier as listed below. Instance names may be up to 1024 characters long.

Table 5: Element Identifiers

Key Letter (First Char)	Element	Example Line
C	Capacitor	Cbypass 1 0 10pf
E	Voltage-controlled voltage source	Ea 1 2 3 4 1K
F	Current-controlled current source	Fsub n1 n2 vin 2.0
G	Voltage-controlled current source	G12 4 0 3 0 10
H	Current-controlled voltage source	H3 4 5 Vout 2.0

K	Linear mutual inductor (general form)	K1 L1 L2 1
L	Linear inductor	LX a b 1e-9
R	Resistor	R10 21 10 1000
S	S-parameter element	S1 nd1 nd2 s_model2
T	Transmission Line	Txxx in 0 out 0 Zo=50 TD=30n
V	Shorting element (DC 0 V voltage source)	V1 8 0 DC=0
W	Transmission Line	W1 in1 0 out1 0 N=1 L=1
X	Subcircuit instance	X1 2 4 17 31 MULTI WN=100 LN=5

## 4.7 Numbers

Numbers may be entered as integer, floating point, floating point with an integer exponent, or integer or floating point with one of the scale factors listed below.

**Table 6: Scale Factors**

Scale Factor	IEEE Standard Prefix	IEEE Standard Symbol	Multiplying Factor
T	Tera	T	1e+12
G	giga	G	1e+9
MEG	mega	M	1e+6
K	kilo	k	1e+3

MIL	n/a	mil	25.4e-6
M	milli	m	1e-3
U	micro	μ	1e-6
N	nano	n	1e-9
P	pico	p	1e-12
F	Femto	f	1e-15
A	Atto	a	1e-18

**Note:**

Scale factor A is not a scale factor in a character string that contains amps. For example, IBIS-ISS-compliant tools shall interpret the string “20amps” as 20 amperes of current, not as 20e-18mps.

- Numbers may use exponential format or engineering key letter format, but not both (1e-12 or 1p, but not 1e-6u).
- To designate exponents, use D or E.
- Trailing alphabetic characters are interpreted as units comments.
- Units comments are not checked.

## 4.8 Parameters and Expressions

- Parameter names shall begin with an alphabetic character. Subsequent characters in the parameter name shall each be either a digit, or one of these following characters:  
! # \$ % [ ] \_
- If multiple definitions are given for the same parameter, IBIS-ISS uses the last parameter definition even if that definition occurs later in the input than a reference to the parameter.
- A parameter shall be defined before that parameter is used in a definition for another parameter.

- To delimit expressions, single quotes shall be used.
- Expressions shall not exceed 1024 characters.

Parameters are evaluated only once, at the time of parsing. Dynamic, recursive or iterative definitions of parameters are prohibited (e.g., parameters defined in terms of the voltage at a node, where that voltage is evaluated at every time step in a transient analysis).

- Parameters are used in two contexts.
  - Parameters in parameter definition statements are strings, defining names that are assigned specific values by the statement. These values may themselves be interpreted as strings (using the “str()” construction noted elsewhere), numeric values, an expression or equation, or strings matching parameters defined elsewhere.
  - Parameters may also appear in element instances, model definitions and subcircuit definitions. These parameters may be user-defined or may use names pre-defined by the syntax of the element. Parameter names are input arguments. Delimiters shall precede and follow names.
- Parameter names may be up to 1024 characters long and are not case-sensitive.

#### **4.9 Node Name (or Node Identifier) Conventions**

Nodes are the points of connection between elements in the input netlist. Only alphanumeric strings shall be used to designate nodes. If entirely numeric, node numbers shall be between 1 and 999999999999999 (1 to 1e16-1). A node number of 0 is permitted but is interpreted as ground. Letters that follow a leading number in a node name are ignored; this means that node strings such as ‘3n5’ and ‘3’ shall be interpreted as referring to the same node.

When the node name begins with a letter or a valid special character, the node name may contain a maximum of 1024 characters.

To indicate the ground node, use either the number 0, the name GND, or !GND, or GROUND, GND!. Every node shall have at least two connections, except for transmission line nodes (unterminated transmission lines are permitted).

#### **4.10 Element, Instance, and Subcircuit Naming Conventions**

Instances and subcircuits are elements and as such, follow the naming conventions for elements.

Element names begin with an identifying letter designating the element type, followed by up to 1023 alphanumeric characters. Element identifiers are R for resistor, C for capacitor and so on.

Line Continuations

Input may be continued across a line-termination sequence in one of three ways:

- Statements are continued across line-termination sequences using the + character
- Quoted strings are continued across line-termination sequences using the \\ sequence
- Individual arguments are continued using the \\ sequence, without leading whitespace

To continue a statement across multiple lines, the plus (+) sign shall be used as the first non-numeric, non-blank character of each continued line. The + sign shall be used only between arguments and delimiters and never to split up the character sequence used for a single argument.

Here is an example of comments and line continuation in a netlist file:

```
*This shows continuation of a statement describing a resistor  
Rexample  
+ n3 n4 R=30
```

To continue a statement with extended length across a line-termination sequence, an argument may be split using a double backslash ( \\ ) sequence. Leading whitespace shall not be included before the double backslash sequence. A quoted string (such as an expression) may be continued across a line-termination sequence using a double backslash sequence with or without leading whitespace.

Here are examples of string and argument continuation.

```
*** string continuation ***  
R6 4 0 R='res1-\\  
res2'  
R5 4 0 R='res1- \\  
res2'  
*** argument continuation ***  
R4 node1 no\\  
de2 R= 'res1-res2'
```

## 5 Parameters

Parameters are similar to the variables used in most programming languages. Parameters hold values assigned when the circuit design is created or that are calculated based on circuit solution values. Parameters may store static values for a variety of quantities (resistance, source voltage, rise time, and so on). Parameters may also be alphabetic strings used with elements where string input is expected (for example, filenames or model names).

### 5.1 Using Parameters in Simulation (.PARAM)

#### Defining Parameters

Parameters may be defined using the methods shown below. Note that a .param statement without an assignment is not permitted.

Table 7: .PARAM Statement Syntax and Examples

Usage	Description/Example
Simple Assignment	<code>.PARAM SimpleParam=1e-12</code>
Algebraic Definition	<code>.PARAM AlgebraicParam='SimpleParam*8.2'</code>
User-defined Function	<code>.PARAM MyFunc(x,y)='SQRT((x*x)+(y*y))'</code>
String Assignment	<code>.PARAM StringParam=STR('mystring')</code>
Subcircuit Definition	<code>.SUBCKT SubName ParamDefName=Value</code>
Subcircuit Instance	<code>Xxxx nodename1 ... nodenamen + SubName + ParamDefName = Value   STR('string')</code>
Predefined Analysis Function	<code>.PARAM f(a,b)=POW(a,2)+a*b g(d)=SQRT(d) + h(e)=e*f(1,2)-g(3)</code>

A parameter definition in IBIS-ISS always uses the last value found in the input netlist. The definitions below assign a value of 3 to the `DupParam` parameter.

```
.PARAM DupParam=1
...
.PARAM DupParam=3
```

IBIS-ISS assigns 3 as the value for all instances of `DupParam`, including instances that are earlier in the input than the `.PARAM DupParam=3` statement.

Note that any ordered substring of “.parameters” containing at least the characters “.para” is also acceptable as an alternative to “.param”.

The parameter resolution order is:

1. Resolve all literal assignments.
2. Resolve all expressions.
3. Resolve all function calls.

Table 8 shows the parameter passing order.

**Table 8: Parameter Passing Order**

.PARAM statement ()	.SUBCKT call (instance)
.SUBCKT call (instance)	.SUBCKT definition (symbol)
.SUBCKT definition (symbol)	.PARAM statement ()

## Assigning Parameters

The following types of values may be assigned to parameters:

- Constant real number
- Algebraic expression of real values
- Predefined function
- Strings not for algebraic evaluation

Any complex expression to be evaluated shall be enclosed in single quotes.

A simple expression consists of one parameter name. Simple expressions shall not be enclosed in single or double quotes.

The parameter keeps the assigned value, unless a later definition changes its value.

## 5.2 Using Algebraic Expressions

In IBIS-ISS, an algebraic expression, with quoted strings, may replace any parameter.

Some uses of algebraic expressions are:

- Parameters:

```
.PARAM x='y+3'
```

- Inline algebra in elements:

```
R1 1 0 r='27*3.14'
```

As noted above, the continuation character sequence in IBIS-ISS is a double backslash (`\\`). For quoted strings, the double backslash may be preceded by whitespace or used without whitespace.

## Built-In Functions and Variables

In addition to simple arithmetic operations (+, -, \*, /), the built-in functions and variables listed below may be used in IBIS-ISS expressions.

Table 9: IBIS-ISS Built-in Functions

IBIS-ISS Form	Function	Class	Description
sin(x)	sine	trig	Returns the sine of x (radians)
cos(x)	cosine	trig	Returns the cosine of x (radians)
tan(x)	tangent	trig	Returns the tangent of x (radians)
asin(x)	arc sine	trig	Returns the inverse sine of x (radians)
acos(x)	arc cosine	trig	Returns the inverse cosine of x (radians)

atan(x)	arc tangent	trig	Returns the inverse tangent of x (radians)
sinh(x)	hyperbolic sine	trig	Returns the hyperbolic sine of x (radians)
cosh(x)	hyperbolic cosine	trig	Returns the hyperbolic cosine of x (radians)
tanh(x)	hyperbolic tangent	trig	Returns the hyperbolic tangent of x (radians)
abs(x)	absolute value	math	Returns the absolute value of x: $ x $
sqrt(x)	square root	math	Returns the square root of the absolute value of x: $\text{sqrt}(-x)=-\text{sqrt}( x )$
pow(x,y)	absolute power	math	Returns the value of x raised to the integer part of y: $x^{(\text{integer part of } y)}$
pwr(x,y)	signed power	math	Returns the absolute value of x, raised to the y power, with the sign of x: $(\text{sign of } x) x ^y$
x**y	power		If $x < 0$ , returns the value of x raised to the integer part of y. If $x = 0$ , returns 0. If $x > 0$ , returns the value of x raised to the y power.
log(x)	natural logarithm	math	Returns the natural logarithm of the absolute value of x, with the sign of x: $(\text{sign of } x)\log( x )$
log10(x)	base 10 logarithm	math	Returns the base 10 logarithm of the absolute value of x, with the sign of x: $(\text{sign of } x)\log_{10}( x )$

exp(x)	exponential	math	Returns e, raised to the power x: $e^x$
db(x)	decibels	math	Returns the base 10 logarithm of the absolute value of x, multiplied by 20, with the sign of x: $(\text{sign of } x)20\log_{10}( x )$
int(x)	integer	math	Returns the integer portion of x. The fractional portion of the number is lost.
nint(x)	integer	math	Rounds x up or down, to the nearest integer.
sgn(x)	return sign	math	Returns -1 if x is less than 0. Returns 0 if x is equal to 0. Returns 1 if x is greater than 0
sign(x,y)	transfer sign	math	Returns the absolute value of x, with the sign of y: $(\text{sign of } y) x $
def(x)	parameter defined	control	Returns 1 if parameter x is defined. Returns 0 if parameter x is not defined.
min(x,y)	smaller of two args	control	Returns the numeric minimum of x and y
max(x,y)	larger of two args	control	Returns the numeric maximum of x and y
[cond] ?x : y	ternary operator		Returns x if <i>cond</i> is not zero. Otherwise, returns y. .param z='condition ? x:y'
<	relational operator (less than)		Returns 1 if the left operand is less than the right operand. Otherwise, returns 0. .param x=y<z (y less than z)

<=	relational operator (less than or equal)		Returns 1 if the left operand is less than or equal to the right operand. Otherwise, returns 0. .param x=y<=z (y less than or equal to z)
>	relational operator (greater than)		Returns 1 if the left operand is greater than the right operand. Otherwise, returns 0. .param x=y>z (y greater than z)
>=	relational operator (greater than or equal)		Returns 1 if the left operand is greater than or equal to the right operand. Otherwise, returns 0. .param x=y>=z (y greater than or equal to z)
==	equality		Returns 1 if the operands are equal. Otherwise, returns 0. .param x=y==z (y equal to z)
!=	inequality		Returns 1 if the operands are not equal. Otherwise, returns 0. .param x=y!=z (y not equal to z)
&&	Logical AND		Returns 1 if neither operand is zero. Otherwise, returns 0. .param x=y&&z (y AND z)
	Logical OR		Returns 1 if either or both operands are not zero. Returns 0 only if both operands are zero. .param x=y  z (y OR z)

Table 10: IBIS-ISS Special Variables

IBIS-ISS Form	Function	Class	Description
---------------	----------	-------	-------------

Time	current simulation time	control	Uses parameters to define the current simulation time, during transient analysis.
Temper	current circuit temperature	control	Uses parameters to define the current simulation temperature, during transient/temperature analysis.
Hertz	current simulation frequency	control	Uses parameters to define the frequency, during AC analysis.

### 5.3 String Parameters

Parameters may be defined and instantiated using strings. String parameters must use special syntax; characters such as single quotes ('), double quotes ("), or curly brackets ({} ) alone are not sufficient for string parameter definition or instantiation.

When defining a parameter that is a character string, the combination `str('string')` shall be used to define the parameter, where *string* is the string to be used as the parameter value. When an instance of the parameter is used, the parameter name is called as `str(parameter_name)`.

IBIS-ISS supports string parameter definition and instantiation for the following netlist components:

- .PARAM statements
- .SUBCKT statements
- TSTONEFILE keyword in the S-element
- FILE and MODEL arguments
- W-element arguments RLGCMODEL and TABLEMODEL

### 5.4 Parameter Scoping and Passing

A parameter is defined either by a `.param` statement (local to that subcircuit), or may be passed into a subcircuit, or may be defined on a `.subckt` definition line.

All parameters defined within a subcircuit are local to that subcircuit under the defined name. Parameter values may be passed between subcircuits, but the values must be passed explicitly, with the parameter present both in the subcircuit definition and in the instance where it is used. IBIS-ISS does not support global parameters.

```
.param x=0
.subckt def
.param x=1
x1 1 2 abc x=2
.subckt abc 1 2 x=3
.param x=3
r1 1 2 R=x
.ends abc
.ends def
.end
```

## 6 File Includes

The include statement inserts another file's contents in the current file at evaluation. Note that any ordered substring of ".include" containing at least the characters ".inc" is also acceptable as an alternative to ".include".

### Syntax

```
.INCLUDE `file_path file_name`
```

```
.inc `file_path file_name`
```

Table 11: File Include Arguments

Argument	Description
<i>file_path</i>	Path name of a file for computer operating systems that support tree-structured directories. An include file can contain nested .INCLUDE calls to another include file.
<i>file_name</i>	Name of a file to include in the data file. Any name valid under the computer's operating system may be used.

Use this statement to include another netlist in the current circuit description. The file path and file name shall be enclosed in single or double quotation marks.

```
.INCLUDE `~/myhome/subcircuits/circuit`
```

## 7 Comments

Comments require an asterisk (\*) as the first character in a line or a dollar sign (\$) directly in front of the comment anywhere on the line.

### Syntax

```
* <comment_on_a_line_by_itself>
```

or

```
<IBIS-ISS statement> $ <comment following input>
```

Comment statements may appear anywhere in the circuit description. The dollar sign (\$) shall be used for comments that do *not* begin in the first character position on a line (for example, for comments that follow simulator input on the same line). If it is not the first nonblank character, then the dollar sign shall be preceded by either:

- Whitespace
- Comma (,)
- Valid numeric expression

The dollar sign may also be used within node or element names. For example:

```
* RF=1K GAIN SHOULD BE 100
$ CIRCUIT EXAMPLE
VIN 1 0 PL 0 0 5V 5NS $ 10v 50ns
R12 1 0 1MEG $ FEED BACK
.PARAM a=1w$comment a=1, w treated as a space and ignored
.PARAM a=1k$comment a=1e3, k is a scale factor
```

A dollar sign is the preferred way to indicate comments, because of the flexibility of its placement within the code.

Comments shall not appear after the double-slash (\\) line continuation sequence on the same line (before a line termination sequence).

## **8 Model Definitions (.MODEL Statements)**

Model definitions are used to specify the electrical parameters for W-element and S-element instances. They can be considered a special form of subcircuit definition, in which the defined subcircuit is only available to W- and S-elements.

The specific syntax for W-element and S-element .MODEL definitions are detailed below, as part of the W-element and S-element portions of the IBIS-ISS specification. Note that .MODEL statements are hierarchically at the same level as element instances.

## 9 Subcircuit Definitions

A subcircuit definition groups a set of statements into a module that may be reused an unlimited number of times. The definition is instantiated using the X-element.

### Syntax

```
.subckt name n1 [n2 n3 ...]  
statement  
statement  
statement  
...  
.ends
```

### 9.1 Subcircuit Scoping Rules

A .subckt or .model definition shall occur in the subcircuit in which the subcircuit or model is referenced, or in a calling subcircuit at any level above.

## 10 Subcircuit Definition Ending Statements

Subcircuit definitions shall end with the `.ends` statement. See Subcircuit Definitions above for syntax and examples.

# 11 Elements

The sections below describe the individual circuit elements that may appear in an IBIS-ISS file.

## 11.1 Subcircuits

Reusable cells are the key to saving labor in any CAD system. To create and simulate a reusable circuit, construct it as a subcircuit. Parameters are used to expand the utility of a subcircuit.

`X<subcircuit_name>` creates an instance of a subcircuit. . The subcircuit shall have already been defined elsewhere in the IBIS-ISS file using a `.SUBCKT` statement.

### Syntax

```
Xxxxx n1 [n2 n3 ...] subnam  
[parnam = val] [M = val]
```

The name of an X-element instance shall begin with the character “X”, followed by up to 1023 alphanumeric characters.

Table 12: Subcircuit Arguments

Argument	Definition
<i>n1 ...</i>	Node names for external reference.
<i>subnam</i>	Subcircuit model reference name.
<i>parnam=val</i>	A parameter name ( <i>parnam</i> ) set to a value ( <i>val</i> ) for use only in the subcircuit. It overrides a parameter value in the subcircuit definition, but is overridden by a value set in a <code>.PARAM</code> statement.
<b>M=val</b>	Multiplier value <i>val</i> instantiates <i>val</i> number of subcircuits connected in parallel to the named nodes. The multiplier <i>val</i> shall be an integer.

## 11.2 Linear Resistor

A linear resistor is a basic electrical circuit element for impeding current flow.

## Syntax

`Rxxx node1 node2 [R =] value`

The value of a linear resistor may be a constant, or an expression of parameters.

The name of an R-element instance shall begin with the character “R”, followed by up to 1023 alphanumeric characters.

Table 13: Linear Resistor Arguments

Argument	Description
<i>node1, node2</i>	Names or numbers of the connecting nodes
<b>R=</b> <i>value</i>	Resistance value, in ohms. The text “R=” is optional.

## 11.3 Linear Capacitor

A linear capacitor is a basic electrical circuit element for charge storage.

### Syntax

`Cxxx node1 node2 [C =] value`

The value of a linear capacitor may be a constant, or an expression of parameters.

The name of a C-element instance shall begin with the character “C”, followed by up to 1023 alphanumeric characters.

Table 14: Linear Capacitor Arguments

Argument	Description
<i>node1, node2</i>	Names of connecting nodes.
<b>C=</b> <i>value</i>	Capacitance value, in farads. The text “C=” is optional.

## 11.4 Voltage Source

A voltage source establishes a DC potential difference between two nodes. A zero-volt source may be used to create a short between two nodes, for use as a current probe, for example.

### Syntax

```
Vxxx node1 node2 [DC =] val
```

The name of a V-element instance shall begin with the character “V”, followed by up to 1023 alphanumeric characters.

Table 15: Voltage Shunt Arguments

Argument	Description
<i>node1, node2</i>	Nodes between which the shunt is placed.
<b>DC=</b> <i>val</i>	Sets the integer <i>val</i> as the voltage between the source nodes. The text “ <b>DC=</b> ” is optional.

## 11.5 Mutual Inductor

A mutual inductor describes inductive coupling between two defined inductors.

### Syntax

```
Kxxx Lyyy Lzzz [K =] coupling
```

The name of a K-element instance shall begin with the character “K”, followed by up to 1023 alphanumeric characters.

Table 16: Mutual Inductor Arguments

Argument	Description
<b>Lyyy</b>	Name of the first of two coupled inductors. This inductor shall be defined elsewhere in the file.
<b>Lzzz</b>	Name of the second of two coupled inductors. This inductor

	shall be defined elsewhere in the file.
<b>K=</b> <i>coupling</i>	Sets <i>coupling</i> as the coefficient of mutual coupling. This is a unitless number, with magnitude > 0. If the coupling coefficient is negative, the direction of coupling reverses. This is equivalent to reversing the polarity of either of the coupled inductors. Use the <b>K=xxx</b> syntax when defining the coupling coefficient using a parameter name or an equation. The text “ <b>K=</b> ” is optional.

## 11.6 Linear Inductor

### Syntax

**Lxxx** *node1 node2* [**L =**] *inductance*

The name of an L-element instance shall begin with the character “L”, followed by up to 1023 alphanumeric characters.

Table 17: Linear Inductor Arguments

Argument	Description
<i>node1, node2</i>	Names or numbers of the connecting nodes.
<b>L=</b> <i>inductance</i>	Sets <i>inductance</i> as the inductance value, in henries. The text “ <b>L=</b> ” is optional.

## 11.7 T-element (Ideal Transmission Line)

### Syntax

**Txxx** *in refin out refout* **Zo=***val* **TD=***val* [**L=***val*]

The name of a T-element instance shall begin with the character “T”, followed by up to 1023 alphanumeric characters.

Table 18: T-element (Ideal Transmission Line) Arguments

Argument	Description
<i>In</i>	Signal input node.
<i>refin</i>	Ground reference for the input signal.
<i>out</i>	Signal output node.
<i>refout</i>	Ground reference for the output signal.
<b>Zo=</b> <i>val</i>	Sets <i>val</i> as the characteristic impedance of the transmission line in ohms. Note that the beginning character sequence may be either <i>zo</i> or <i>z0</i> .
<b>TD=</b> <i>val</i>	Sets <i>val</i> as the propagation time delay of the transmission line, in seconds. If physical length (L) is specified, then units for TD are considered in seconds per meter.
<b>L=</b> <i>val</i>	Sets <i>val</i> as the physical length of the transmission line, in meters. Default=1.

## 11.8 W-element (Coupled Transmission Line)

This section describes how to use basic transmission line simulation equations and an optional method for computing the parameters of transmission line equations.

The W-element is a versatile transmission line model that may be used to describe a variety of transmission line structures, from a simple lossless line to complex frequency-dependent lossy-coupled lines.

### Syntax

```
Wxxx i1 i2 ... iN iR o1 o2 ... oN oR N=val L=val
+ [RLGCMODEL=name | TABLEMODEL=name]
```

The name of a W-element instance shall begin with the character “W”, followed by up to 1023 alphanumeric characters.

Table 19: W-element (Coupled Transmission Line) Arguments

Argument	Description
<b>N=val</b>	Sets non-zero integer <i>val</i> as the number of signal conductors (excluding the reference conductor).
<i>i1...iN</i>	Node names for the near-end signal-conductor terminal
<i>iR</i>	Node name for the near-end reference-conductor terminal
<i>o1... oN</i>	Node names for the far-end signal-conductor terminal
<i>oR</i>	Node name for the far-end reference-conductor terminal
<b>L=val</b>	Defines <i>val</i> as the length of the transmission line, in meters
<b>RLGCMODEL=name</b>	Defines string <i>name</i> as the name of the RLGC model

<b>TABLEMODEL=<i>name</i></b>	Defines string <i>name</i> as the name of the frequency-dependent tabular model
-------------------------------	---------------------------------------------------------------------------------

The W-element supports two formats to specify transmission line properties:

- Format 1: RLGC specification
  - Internally specified in a `.MODEL` statement.
  - Externally specified in a different file.
- Format 2: Frequency-dependent tabular specification

Parameters in the W-element element declaration may be declared in any order. Specify the number of signal conductors, N, after the list of nodes. The nodes and parameters in the W-element element declaration may be interspersed.

## Format 1: RLGC Model

The inputs of the W-element are given in per unit length matrices:  $R_o$  (DC resistance), L, G, C,  $R_s$  (skin effect), and  $G_d$  (dielectric loss)

The W-element does not limit any of the following parameters:

- Number of coupled conductors.
- Shape of the matrices.
- Line loss.
- Length or amount of frequency dependence.

The RLGC text file contains frequency-dependent RLGC matrices per unit length. The W-element also handles frequency-independent RLGC, and lossless (LC) lines. It does not support RC lines.

Because RLGC matrices are symmetrical, the RLGC model specifies only the lower triangular parts of the matrices. The syntax of the RLGC model for the W-element is:

```
.MODEL name W MODELTYPE=RLGC N=val
+ Lo=matrix_entries
+ Co=matrix_entries [Ro=matrix_entries Go=matrix_entries]
+ Rs=matrix_entries wp=val Gd=matrix_entries Rognd=val
+ Rsgnd=val Lgnd=val
```

Table 20: W-element RLGC Model Arguments

Argument	Description
<b>N=</b> <i>val</i>	Number of conductors (same as in the element card).
<b>L=</b> <i>matrix_entries</i>	DC inductance matrix, per unit length $\left[\frac{\text{H}}{\text{m}}\right]$ .
<b>C=</b> <i>matrix_entries</i>	DC capacitance matrix, per unit length $\left[\frac{\text{F}}{\text{m}}\right]$ .
<b>Ro=</b> <i>matrix_entries</i>	DC resistance matrix, per unit length $\left[\frac{\Omega}{\text{m}}\right]$ .
<b>Go=</b> <i>matrix_entries</i>	DC shunt conductance matrix, per unit length $\left[\frac{\text{S}}{\text{m}}\right]$ .
<b>Rs=</b> <i>matrix_entries</i>	Skin effect resistance matrix, per unit length $\left[\frac{\Omega}{\text{m}\sqrt{\text{Hz}}}\right]$ .
<b>Gd=</b> <i>matrix_entries</i>	Dielectric loss conductance matrix, per unit length $\left[\frac{\text{S}}{\text{m}\cdot\text{Hz}}\right]$ .
<b>wp</b>	Angular frequency of the polarization constant [radian/sec] (see Introduction to the Complex Dielectric Loss Model). When the wp value is specified, the unit of Gd becomes [S/m].
<b>Lgnd=</b> <i>val</i>	Defines <i>val</i> as the DC inductance value, per unit length for grounds $\left[\frac{\text{H}}{\text{m}}\right]$ (reference line).
<b>Rognd=</b> <i>val</i>	Defines <i>val</i> as the DC resistance value, per unit length for

	ground $\left[ \frac{\Omega}{\text{m}} \right]$ .
<b>Rsgnd=val</b>	Defines <i>val</i> as the skin effect resistance value, per unit length for ground $\left[ \frac{\Omega}{\text{m}\sqrt{\text{Hz}}} \right]$ .

The following input netlist file shows RLGC input for the W-element:

```
* W-Element example, four-conductor line
W1 N=3 1 3 5 0 2 4 6 0 RLGCMODEL=example_rlc l=0.97

* RLGC matrices for a four-conductor lossy
.MODEL example_rlc W MODELTYPE=RLGC N=3
+ Lo=
+ 2.311e-6
+ 4.14e-7 2.988e-6
+ 8.42e-8 5.27e-7 2.813e-6
+ Co=
+ 2.392e-11
+ -5.41e-12 2.123e-11
+ -1.08e-12 -5.72e-12 2.447e-11
+ Ro=
+ 42.5
+ 0 41.0 + 0 0 33.5
+ Go=
+ 0.000609
+ -0.0001419 0.000599
+ -0.00002323 -0.00009 0.000502
+ Rs=
+ 0.00135
+ 0 0.001303
+ 0 0 0.001064
+ Gd=
+ 5.242e-13
+ -1.221e-13 5.164e-13
+ -1.999e-14 -7.747e-14 4.321e-13
```

## Using RLGC Matrices

RLGC matrices in the RLGC model of the W-element are in the Maxwellian format.

## Format 2: Frequency-Dependent Tabular Specification

The tabular RLGC model may be used as an extension of the analytical RLGC model to model any arbitrary frequency-dependent behavior of transmission lines (this model does not support RC lines).

The W-element syntax supports tables of data (use a `.MODEL` statement of type `w`). To accomplish this, the `.MODEL` statement refers to `.MODEL` statements where the “type” is `SP` (described in Small-Signal Parameter Data Frequency Table Model (SP Model)), which contain the actual table data for the RLGC matrices.

To ensure accuracy, the W-element tabular model requires the following:

- R and G tables shall include zero-frequency data points.
- L and C tables shall include infinite-frequency data points as well as zero-frequency data points.

To specify a zero-frequency point, either the `DC` argument shall be used or, alternatively, the `f` parameter in the `DATA` field of the `SP` model may be set to a value of 0. To specify an infinity frequency point, use the `INFINITY` argument.

See also, Small-Signal Parameter Data Frequency Table Model (SP Model).

## 11.9 Frequency-Dependent Matrices

The static (constant) L and C matrices are accurate for a wide range of frequencies. In contrast, the static (DC) R matrix applies to only a limited frequency range, mainly due to the skin effect. A good approximate expression of the R resistance matrix with the skin effect, is:

Equation 1

$$R(f) \cong R_o + \sqrt{f}(1 + j)R_s$$

where:

- $R_o$  is the DC resistance matrix.
- $R_s$  is the skin effect matrix.

The imaginary term depicts the correct frequency response at high frequency; however, it might cause significant errors for low-frequency applications. In the W-element, this imaginary term may be excluded:

```
Wxxx i1 i2 ... iN iR o1 o2 ... oN oR N=val L=val INCLUDERSIMAG=NO
```

In contrast, the G (loss) conductance matrix is often approximated as:

Equation 2

$$G(f) \cong G_o + \frac{f}{\sqrt{1 + \left(\frac{f}{f_{gd}}\right)^2}} G_d$$

Where,

- $G_o$  models the shunt current due to free electrons in imperfect dielectrics.

- $G_d$  models the power loss due to the rotation of dipoles under the alternating field.
- $f_{gd}$  is a cut-off frequency.

If  $f_{gd}$  is not set, or if  $f_{gd}$  is set to 0, then  $G(f)$  keeps linear dependency on the frequency. In the  $W$ -element, the default  $f_{gd}$  is zero (that is,  $G(f)$  does not use the  $f_{gd}$  value).

An alternate value may be included in the  $W$ -element statement:

```
Wxxx i1 i2 ... iN iR o1 o2 ... oN oR N=val L=val fgd=val
```

To use the previous linear dependency, set  $f_{gd}$  to 0.

**Note:**  $f_{gd}$  shall be used to estimate frequency dependent shunt loss conductance for the RLGC model when INCLUDEGDIMAG=yes has not been set.

When INCLUDEGDIMAG=yes, the RLGC model shall estimate frequency-dependent shunt (C and G) parameters described as Equation 2 and the  $f_{gd}$  value shall be ignored.

Both of these are ways to fit the RLGC model fit with actual measurements. If measured or computationally extracted data is used for a tabular RLGC model, it should be more accurate if parameter extraction is accurately done.

## Small-Signal Parameter Data Frequency Table Model (SP Model)

The small-signal parameter data frequency table model (SP model) is a generic model that describes frequency-varying behavior.

### Syntax

```
.MODEL name sp [N=val FSTART=val FSTOP=val NI=val
+ SPACING=spacing MATRIX=matrix VALTYPE=valtype
+ INFINITY=matrixval INTERPOLATION=val EXTRAPOLATION=val]
+ [DATA=(npts ...) | DATAFILE=filename]
```

### Parameter Description

Table 21:  $W$ -element SP Model Arguments

Argument	Description
----------	-------------

<b>N=</b> <i>val</i>	Sets non-zero integer <i>val</i> as the Matrix dimension (number of signal terminals). Default is 1.
<b>FSTART=</b> <i>val</i>	Sets <i>val</i> as the starting frequency point for data. Default=0.
<b>FSTOP=</b> <i>val</i>	Sets <i>val</i> as the final frequency point for data. Use this parameter only for the LINEAR and LOG spacing formats.
<b>NI=</b> <i>val</i>	Number of frequency points per interval. Use this parameter only for the DEC and OCT spacing formats. Default=10.
<b>SPACING=</b> <i>spacing</i>	Sets <i>spacing</i> as one of the following data sample spacing formats: <ul style="list-style-type: none"> <li>• LIN (LINEAR): uniform spacing with frequency step of (FSTOP-FSTART)/(npts-1). The default.</li> <li>• OCT: octave variation with FSTART as the starting frequency and NI points per octave. npts sets the final frequency.</li> <li>• DEC: decade variation with FSTART as the starting frequency and NI points per decade. npts sets the final frequency.</li> <li>• LOG: logarithmic spacing. FSTART and FSTOP are the starting and final frequencies.</li> <li>• POI: non-uniform spacing. Pairs data (NONUNIFORM) points with frequency points.</li> </ul>
<b>MATRIX=</b> <i>matrix</i>	Sets <i>matrix</i> as one of the following matrix (data point) formats: <ul style="list-style-type: none"> <li>• SYMMETRIC: symmetric matrix. Specifies only lower-half triangle of a matrix (default).</li> <li>• HERMITIAN: similar to SYMMETRIC; off-diagonal terms are complex conjugates of each other.</li> <li>• NONSYMMETRIC: non-symmetric (full) matrix.</li> </ul>
<b>VALTYPE=</b> <i>valtype</i>	Sets <i>valtype</i> as one of following data types for matrix elements: <ul style="list-style-type: none"> <li>• REAL: real entry.</li> </ul>

	<ul style="list-style-type: none"> <li>• CARTESIAN: complex number in real/imaginary format (default).</li> <li>• POLAR: complex number in polar format. Specify angles in radians.</li> </ul>
<b>INFINITY</b> = <i>matrixval</i>	Sets <i>matrixval</i> as the data point at infinity. Typically real-valued. This data format shall be consistent with the MATRIX and VALTYPE specifications. Npts does not count this point.
<b>INTERPOLATION</b> = <i>val</i>	Sets <i>val</i> as the interpolation scheme: STEP: piecewise step (default). LINEAR: piecewise linear. SPLINE: b-spline curve fit.
<b>EXTRAPOLATION</b> = <i>val</i>	<p>Sets <i>val</i> as one of the following the extrapolation schemes to be used for the model data during simulation:</p> <ul style="list-style-type: none"> <li>• NONE: no extrapolation is allowed. Simulation terminates if a required data point is outside of the specified range.</li> <li>• STEP: uses the last boundary point. The default.</li> <li>• LINEAR: linear extrapolation by using the last two boundary points.</li> </ul> <p>If the data point at infinity is specified, then extrapolation is not used.</p>
<i>npts</i>	Number of data points.
<b>DATA</b> = <i>filename</i>	<p>Data points. The syntax to be used depends on spacing:</p> <ul style="list-style-type: none"> <li>• Syntax for LIN spacing: .MODEL name sp SPACING=LIN [N=dim] FSTART=f0 + DF=f1 DATA=npts d1 d2 ...</li> <li>• Syntax for OCT or DEC spacing: .MODEL name sp SPACING=DEC or OCT [N=dim] + FSTART=f0 NI=n_per_intval DATA=npts d1 d2 ...</li> <li>• Syntax for POI spacing: .MODEL name sp SPACING=NONUNIFORM [N=dim] + DATA=npts f1 d1 f2 d2 ...</li> </ul>

<b>DATAFILE=filename</b>	Sets filename as the name of an external file containing the data points
--------------------------	--------------------------------------------------------------------------

If an external file is used for the SP model data points, this file must contain only raw numbers without any suffixes, comments or continuation characters. The first number in the file must be an integer value to indicate the number of sampling points in the file (*npts*). Then, sampling data must follow. The order of sampling data must be the same as in the DATA statement. This data file has no limitation on line length.

## W-element Model Definition Syntax

```
.MODEL name W MODELTYPE=TABLE [FITGC=0|1] N=val
+ LMODEL=l_freq_model CMODEL=c_freq_model
+ [RMODEL=r_freq_model GMODEL=g_freq_model]
```

Table 22: W-element Model Definition Arguments

Argument	Description
<b>FITCG=0/1</b>	A value of 1 instructs the tool to run a causality check on the data. A value of 0 turns any causality checking off (default)
<b>N=val</b>	Sets <i>val</i> as the integer, non-zero number of signal conductors (excluding the reference conductor).
<b>LMODEL=l_freq_model</b>	Sets <i>l_freq_model</i> as the SP model name for the inductance matrix array.
<b>CMODEL=c_freq_model</b>	Sets <i>c_freq_model</i> as the SP model name for the capacitance matrix array.
<b>RLMODEL=r_freq_model</b>	Sets <i>r_freq_model</i> as the SP model name for the resistance matrix array. By default, it is zero.
<b>GMODEL=g_freq_model</b>	Sets <i>g_freq_model</i> as the SP model name for the conductance matrix array. By default, it is zero.

## 11.10S-element

An S-element is a frequency-domain set of network data, described using scattering parameters.

### Syntax

```
Sxxx nd1 nd2 ... ndN [ndRef]
+ MNAME=Smodel_name
+ [FBASE = base_frequency] [FMAX = maximum_frequency]
```

The name of an S-element instance shall begin with the character “S”, followed by up to 1023 alphanumeric characters.

Table 23: S-element Arguments

Argument	Description
<i>nd1 nd2...ndN</i>	<p>Nodes of an S-element. Three kinds of definitions are permitted:</p> <ul style="list-style-type: none"> <li>■ With no reference node <i>ndRef</i>, the default reference node is GND. Each node <i>ndi</i> (<i>i</i>=1~<i>N</i>) and GND construct one of the <i>N</i> ports of the S-element.</li> <li>■ With one reference node, <i>ndRef</i> is defined. Each node <i>ndi</i> (<i>i</i>=1~<i>N</i>) and the <i>ndRef</i> construct one of the <i>N</i> ports of the S-element.</li> <li>■ With an <i>N</i> reference node, each port has its own reference node. The node definition may be written more clearly:  <i>nd1+ nd1- nd2+ nd2- ... ndN+ ndN-</i>                      Each pair of the nodes (<i>ndi+</i> and <i>ndi-</i>, <i>i</i>=1~<i>N</i>) constructs one of the <i>N</i> ports of the S-element.</li> </ul>
<i>ndRef</i>	Reference node
<b>MNAME</b> = <i>Smodel_name</i>	Sets <i>Smodel_name</i> as the name of the S model. Note that string parameters are supported in calling an MNAME.

<b>FBASE</b> = <i>base_frequency</i>	<p>Sets <i>base_frequency</i> as the base frequency to use for transient analysis. This value becomes the base frequency point for Inverse Fast Fourier Transformation (IFFT).</p> <ul style="list-style-type: none"> <li>■ If this value is not set, the base frequency is a reciprocal value of the transient period.</li> <li>■ If a frequency is set that is smaller than the reciprocal value of the transient, then transient analysis performs circular convolution, and uses the reciprocal value of FBASE as its base period.</li> </ul>
<b>FMAX</b> = <i>maximum_frequency</i>	<p>Sets <i>maximum_frequency</i> as the maximum frequency used in transient analysis. Used as the maximum frequency point for Inverse Fast Fourier Transformation (IFFT).</p>

The nodes of the S-element shall be placed immediately after the identifier string. All optional parameters in both the S-element and S model statements may be defined by the user, except for MNAME argument.

The optional arguments may be entered in any order, and the parameters specified in the element statement have a higher priority.




Figure 1: Terminal Node Notation

### Node Example

The following example illustrates the  $nd1$   $nd2$ ...  $ndN$  no reference, single reference, and multi-reference parameters.

\*\*S-parameter example

```

* no reference
S_no_ref n1 n2 mname=s_model

* single reference
S_one_ref n1 n3 gnd mname=s_model

*multi-reference
S_multi_ref n1 gnd n4 gnd mname=s_model

```

The S-element must have a call to one of the supported S-parameter file formats (IBIS-ISS gets the number of ports from the S-parameter file The number of ports, 'n', may be specified explicitly as N=n.

- For n terminals, the S-element assumes no reference node.
- For n+1 terminals, the S-element assumes one reference node.
- For 2n terminals, the S-element assumes signal nodes and n reference nodes. Each pair of nodes is a signal and a reference node.

## S-Element Model Syntax

Use the following syntax to describe specific S-element models:

```

.MODEL Smodel_name S [N=dimension]
+ [TSTONEFILE=filename]
+ [FBASE=base_frequency] [FMAX=maximum_frequency]

```

Table 24: S-element Model Definition Arguments

Argument	Description
<i>Smodel_name</i>	Name of the S model.
<b>S</b>	Specifies that the model type is an S model.
<b>N=dimension</b>	Sets non-zero integer <i>dimension</i> as the S model dimension, the number of terminals of the S-element to be defined, excluding the reference node.

<p><b>TSTONEFILE</b>=<i>filename</i></p>	<p>Sets the string <i>filename</i> as the name of a Touchstone file. Data contains frequency-dependent array of matrixes. Touchstone files must follow the .s#p file extension rule, where # represents the dimension of the network.</p> <p>Note that string parameters are supported for TSTONEFILE</p> <p>Example:</p> <pre>.subckt sparam n1 n2 tsfile=str('ss_ts.s2p') S1 n1 n2 0 mname=s_model .model s_model S TSTONEFILE=str(tsfile) .ends x1 A B sparam tsfile=str('ss_ts.s2p') ...</pre> <p>For details, see <i>Touchstone® File Format Specification</i> by the IBIS Open Forum (<a href="http://www.eda.org/ibis/">http://www.eda.org/ibis/</a>).</p>
<p><b>FBASE</b>=<i>base_frequency</i></p>	<p>Sets <i>base_frequency</i> as the expected base frequency to be used for transient analysis involving the associated S-element model. IBIS-ISS uses this value as the base frequency point for Fast Inverse Fourier Transformation (IFFT).</p> <ul style="list-style-type: none"> <li>■ If FBASE is not set, the simulator shall use a reciprocal of the transient period as the base frequency.</li> <li>■ If FBASE is set smaller than the reciprocal value of transient period, transient analysis performs circular convolution by using the reciprocal value of FBASE as a base period.</li> </ul>
<p><b>FMAX</b>=<i>maximum_frequency</i></p>	<p>Sets <i>maximum_frequency</i> as the expected maximum frequency to be used for transient analysis involving the associated S-element model. Used as the maximum frequency point for Inverse Fast Fourier Transform (IFFT).</p>

The TSTONEFILE parameter describes the frequency-varying behavior of a network.

## 11.11 E-element (Voltage-Controlled Voltage Source)

This section explains the E-element syntax and parameters.

### Linear

```
Exxx n+ n- [VCVS] in+ in- gain
```

### Laplace Transform

Voltage Gain H(s):

```
Exxx n+ n- LAPLACE in+ in- k0, k1, ..., kn / d0,  
d1, ..., dm
```

H(s) is a rational function, with parameters used to define the values of all coefficients ( $k_0, k_1, \dots, d_0, d_1, \dots$ ).

### Pole-Zero Function

Voltage Gain H(s):

```
Exxx n+ n- POLE in+ in- a az1, fz1, ..., azn, fzn / b,  
+ ap1, fp1, ..., apm, fpm
```

The following equation defines H(s) in terms of poles and zeros:

$$H(s) = \frac{a \cdot (s + \alpha_{z1} - j2\pi f_{z1}) \dots (s + \alpha_{zn} - j2\pi f_{zn})(s + \alpha_{zn} + j2\pi f_{zn})}{b \cdot (s + \alpha_{p1} - j2\pi f_{p1}) \dots (s + \alpha_{pm} - j2\pi f_{pm})(s + \alpha_{pm} + j2\pi f_{pm})}$$

#### Equation 3

The complex poles or zeros are in conjugate pairs. The element description specifies only one of them, and the program includes the conjugate.

Parameters may be used to specify the a, b,  $\alpha$ , and f values.

#### Example

```
Elow_pass out 0 POLE in 0 1.0 / 1.0, 1.0,0.0 0.5,0.1379
```

The `Elow_pass` statement describes a low-pass filter, with the transfer function:

$$H(s) = \frac{1.0}{1.0 \cdot (s + 1)(s + 0.5 + j2\pi \cdot 0.1379)(s + 0.5 - (j2\pi \cdot 0.1379))}$$

## Foster Pole-Residue Form

Gain E(s) form

```
Exxx n+ n- FOSTER in+ in- k0 k1  
+ (Re{A1}, Im{A1}) / (Re{p1}, Im{p1})  
+ (Re{A2}, Im{A2}) / (Re{p2}, Im{p2})  
+ (Re{A3}, Im{A3}) / (Re{p3}, Im{p3})  
+ ...
```

In the above syntax, parentheses, commas, and slashes are separators—they have the same meaning as a space. A pole-residue pair is represented by four numbers (real and imaginary part of the residue, then real and imaginary part of the pole).

For convergence, the Re[pi] must be less than zero. For an N-port admittance matrix Y, Re{Y} should be positive-definite to ensure passivity of the model.

**Note:**

For real poles, half the residue value is entered because it is applied twice. In the above example, the first pole-residue pair is real, but is written as “A1/(s-p1)+A1/(s-p1)”; therefore, 0.0004 is entered rather than 0.0008.

## E-element Arguments

The E-element arguments are described in the following list. Note that the element name must begin with the character “E” and be followed by up to 1023 alphanumeric characters.

Table 25: E-element Arguments

Argument	Description
<i>gain</i>	Voltage gain
<i>in+/-</i>	Positive or negative controlling nodes. Specify one pair for each dimension
<i>n+/-</i>	Positive or negative node of a controlled element
<b>VCVS</b>	VCVS is a reserved word; do not use it as a node name

## 11.12F-element (Current-Controlled Current Source)

This section explains the F-element syntax and parameters.

### Note:

G-elements with algebraic expressions may be used to duplicate the functions of an F-element.

### Syntax

```
Fxxx n+ n- [CCCS] vn1 gain
```

### F-element Arguments

The F-element parameters are described in the following list. Note that the F-element name must begin with the character “F”, followed by up to 1023 alphanumeric characters.

Table 26: F-element Arguments

Argument	Description
<b>CCCS</b>	Pre-defined argument for current-controlled current source. CCCS is an IBIS-ISS reserved word; do not use it as a node name.
<i>gain</i>	Current gain.
<i>n+/-</i>	Connecting nodes for a positive or negative controlled source
<i>vn1</i>	Names of the voltage source, through which the controlling current flows.

## 11.13G-element (Voltage-Controlled Current Source)

This section explains G-element syntax statements, and their parameters.

### Linear

`Gxxx n+ n- [VCCS] in+ in- transconductance`

For a description of the G-element parameters, see Table VCCS Parameters.

### Laplace Transform

Transconductance H(s):

`Gxxx n+ n- LAPLACE in+ in- k0, k1, ..., kn / d0, d1, ..., dm`

H(s) is a rational function, in the following form:

$$H(s) = \frac{k_0 + k_1s + \dots + k_ns^n}{d_0 + d_1s + \dots + d_ms^m}$$

Equation 4

Parameters may be used to define the values of all coefficients ( $k_0, k_1, \dots, d_0, d_1, \dots$ ).

### Pole-Zero Function

Transconductance H(s):

`Gxxx n+ n- POLE in+ in- a az1, fz1, ..., azn, fzn / b, + ap1, fp1, ..., apm, fpm`

The following equation defines H(s) in terms of poles and zeros:

$$H(s) = \frac{a \cdot (s + \alpha_{z1} - j2\pi f_{z1}) \dots (s + \alpha_{zn} - j2\pi f_{zn})(s + \alpha_{zn} + j2\pi f_{zn})}{b \cdot (s + \alpha_{p1} - j2\pi f_{p1}) \dots (s + \alpha_{pm} - j2\pi f_{pm})(s + \alpha_{pm} + j2\pi f_{pm})}$$

Equation 5

The complex poles or zeros are in conjugate pairs. The element description specifies only one of them, and the simulation program will include the conjugate automatically. Parameters may be used to specify the a, b,  $\alpha$ , and f values.

### Example

Ghigh\_pass 0 out POLE in 0 1.0 0.0,0.0 / 1.0 0.001,0.0

The Ghigh\_pass statement describes a high-pass filter, with the transfer function:

$$H(s) = \frac{1.0 \cdot (s + 0.0 + j \cdot 0.0)}{1.0 \cdot (s + 0.001 + j \cdot 0.0)}$$

## Foster Pole-Residue Form

Transconductance G(s) form

```
Gxxx n+ n- FOSTER in+ in- k0 k1
+ (Re{A1}, Im{A1}) / (Re{p1}, Im{p1})
+ (Re{A2}, Im{A2}) / (Re{p2}, Im{p2})
+ (Re{A3}, Im{A3}) / (Re{p3}, Im{p3})
+ ...
```

In the above syntax, parentheses, commas, and slashes are separators—they have the same meaning as a space. A pole-residue pair is represented by four numbers (real and imaginary part of the residue, then real and imaginary part of the pole).

For convergence, the Re[pi] shall be less than zero. For an N-port admittance matrix Y, Re{Y} should be positive-definite to ensure passivity of the model.

For a description of the G-element parameters, see **Error! Reference source not found.**

### Example

To represent a G(s) in the form,

$$G(s) = 0.001 + 1 \times 10^{-12} s + \frac{0.0008}{s + 1 \times 10^{10}} + \frac{(0.001 - j0.006)}{s - (-1 \times 10^8 + j1.8 \times 10^{10})} + \frac{(0.001 + j0.006)}{s - (-1 \times 10^8 - j1.8 \times 10^{10})}$$

The IBIS-ISS syntax is:

```
G1 1 0 FOSTER 2 0 0.001 1e-12
+(0.0004, 0)/(-1e10, 0) (0.001, -0.006)/(-1e8, 1.8e10)
```

### Note:

For real poles, half the residue value is entered because it is applied twice. In the above example, the first pole-residue pair is real, but is

written as “A1/(s-p1)+A1/(s-p1)”]; therefore, 0.0004 is entered rather than 0.0008.

## G-element Arguments

The G-element arguments are described in the following list. Note that the G-element name must begin with the character “G”, followed by up to 1023 alphanumeric characters.

Table 27: G-element Arguments

Argument	Description
<i>in+/-</i>	Positive or negative controlling nodes. Specify one pair for each dimension.
<i>n+/-</i>	Positive or negative node of the controlled element.
transconductance	Voltage-to-current conversion factor.
<b>VCCS</b>	Pre-defined argument for the voltage-controlled current source. VCCS is a reserved IBIS-ISS word; do not use it as a node name.

### 11.14H-element (Current-Controlled Voltage Source)

This section explains H-element syntax statements, and defines their parameters.

**Note:**

The E-element with algebraic expressions may be used to duplicate the function of the H-element.

**Syntax**

`Hxxx n+ n- [CCVS] vnl transresistance`

Table 28: H-element Arguments

Argument	Description
<b>CCVS</b>	Pre-defined argument for the current-controlled voltage source. CCVS is an IBIS-ISS reserved word; do not use it as a node name.
<i>n+/-</i>	Connecting nodes for positive or negative controlled source.
<i>transresistance</i>	Current-to-voltage conversion factor.
<i>vn1 ...</i>	Names of voltage sources, through which controlling current flows. One name for each dimension shall be specified.

## 12 Best Practices

This section details syntax recommendations for ensuring maximum compatibility with existing proprietary SPICE variants. While not requirements for IBIS-ISS, following these practices will help maintain the portability of IBIS-ISS files.

- Exponent ranges should be limited to between e-60 and e+60.
- The use of a single backslash character (\) as a line continuation character is not supported in IBIS-ISS.
- For maximum compatibility, the “X” character may be used as the “Meg” scale factor. However, while not illegal, this practice should be avoided.
- For maximum compatibility, in-line comments may be preceded by the \* (asterisk) character instead of the \$ (dollar) character. However, while not illegal, this practice should be avoided.
- Instance names shall begin with the appropriate identifying character. The remaining characters of a instance name should be limited to upper- and lower-case alphabetic characters, numeric characters 0-9 and the characters:
  - ~!@#%&\_<>?[]|:;
- Parameter names should begin with [a-z] or [A-Z], and the remaining characters should be limited to upper- and lower-case alphabetic characters, numeric characters 0-9 and the characters:
  - !# \$ % [ ]
- While a parameter may be defined in more than one .param statement within a subcircuit, this practice is best avoided.
- Node names should either be all numeric (0-9), or follow the rules of instance names stated above.

# 13 Theoretical Background

## 13.1 Introduction to the Complex Dielectric Loss Model

When INCLUDEGDIMAG=yes and there is no wp input, the W-element algorithm shall regard the Gd matrix as the conventional model. The tool shall then automatically extract constants for the complex dielectric model.

In conventional use, the W-element RLGC model, frequency-dependent conductance is approximated as Equation 2.

Where, Equation 2 represents the increase of shunt conductance due to dielectric loss. These pure real non-constant functions of frequency violate causality (1). As system operating frequency becomes significantly high even for PCB systems which use high polymer dielectric materials like FR4, the appearance of the dielectric loss becomes significant and significant non-causality of Equation 2 appears.

Frequency-Dependent Matrices

The frequency-dependent loss of the shunt conductance in the dielectric is mainly due to dielectric polarization. This polarization loss leads to a complex permittivity  $\varepsilon(\omega)$  for the dielectric material (2).

Equation 6 
$$\varepsilon(\omega) = \varepsilon'(\omega) - j\varepsilon''(\omega)$$

The loss tangent of the dielectric material can be specified as the ratio of imaginary part of  $\varepsilon(\omega)$  to the real part,

Equation 7 
$$\tan \delta(\omega) = \frac{\varepsilon''(\omega)}{\varepsilon'(\omega)}$$

For a single dielectric dipolar moment, complex electric permittivity can be written as,

Equation 8 
$$\varepsilon(\omega) = \varepsilon_{\infty} + \omega_p \frac{\varepsilon_{dc} - \varepsilon_{\infty}}{j\omega + \omega_p}$$

Where  $\varepsilon_{dc}$ , and  $\varepsilon_{\infty}$  are low and high frequency limits of dielectric permittivity which are real numbers. And  $\omega_p$  is the angular frequency that corresponds to the polarization time constant of the dielectric material. From Equation 5, frequency-dependent complex shunt loss conductance can be expressed as (2)

Equation 9 
$$G(\omega) = G_0 + Gd \frac{j\omega}{j\omega + \omega_p}$$

Where, the imaginary part of the conductance contributes reactively. In cases of multiple dielectric materials surrounding the system, the complex loss conductance can be extended as linear combinations of multiple dipole moments as,

Equation 10 
$$G(\omega) = G_0 + \sum_k Gd_k \frac{j\omega}{j\omega + \omega_{pk}}$$

Since Equation 7 satisfies the Kramers-Kronig condition, we can ensure the passivity/causality of the system. Note that when this new model is activated, the definition of  $G_d$  changes from conventional  $[S/m \cdot Hz]$  to  $[S/m]$ .

## 14 References

1. *Model Order Reduction for Strictly Passive and Causal Distributed Systems*. **Daniel, Luca and Philips, Joel**. 2002. Proceeding of DAC.
2. *Time-Domain Simulation of Multiconductor Transmission Lines with Frequency-Dependent Losses*. **Gordon, Colin, Blazeck, Thomas and Mitra, Raj**. 1992. IEEE Transactions on Computing-Aided Design. Vols. II, No. 11, pp. 1372-1387.
3. *IEEE Transactions on Circuits and Systems-I: Fundamental Theory and Applications*. **Wing, Omar and Yu, Qingjian**. 1994. Vols. 41, No. 2, pp. 107-119.
4. *Rational Approximation of Frequency Domain Responses by Vector Fitting*. **Gustavsen, Bjorn and Semlyen, Adam**. 1999. IEEE Transaction on Power Delivery. Vols. 14, No. 3, pp. 1052-1061.
5. *Accurate models for microstrip computer aided design*. **Hammerstad, E. and Jensen, O**. 1980. IEEE MTT-S Int. Microwave Symp. Dig. pp. 407-409.
6. **Balanis, C.A.** *Advanced Engineering Electromagnetics*. New York : Wiley, 1989.
7. **Bogatin, Eric**. *Signal Integrity Simplified*. Upper Saddle River : Prentice Hall, 2003.
8. *Practical Design Considerations for 10 to 25 Gbps Copper Backplane Serial Links*. **Kollipara, Ravi et al**. Santa Clara : IEC, 2006. DesignCon.